

An overhead photograph of a group of people sitting around a long wooden table in a meeting. The table is cluttered with coffee cups, a smartphone, and some papers. The people are engaged in conversation, with some looking at their phones. The floor has a colorful geometric tile pattern.

# CLUB INFORMATIQUE DE BRETAGNE

Accompagner la transformation digitale



Exemple concret de point d'exit

Envoi d'un message d'alerte lors d'une  
opération sur les profils utilisateur \*ALLOBJ

## Doc IBM

```
dcl-pr MSGUSR ExtPgm('MSGUSR');  
    NewUsrInfo char(38) const;  
end-pr;
```

```
dcl-pi MSGUSR;  
    NewUsrInfo char(38) const;  
end-pi;
```

```
// Declare datastructure for breaking down the  
info  
dcl-ds NewUsrSub;  
    pExtPgm CHAR(20);  
    pExtFmt CHAR(8);  
    pUsrPrf CHAR(10);  
end-ds;
```

```
Select;
When pExtFmt = 'CHGP0100';
  Usage = 'changed';
When pExtFmt = 'CRTP0100';
  Usage = 'created';
When pExtFmt = 'DLTP0100';
  Usage = 'deleted';
When pExtFmt = 'DLTP0200';
  Usage = 'deleted';
When pExtFmt = 'RSTP0200';
  Usage = 'restored';
Endsl;
```

```
EXEC SQL
  SELECT u.authorization_name,
         u.group_profile_name,
         u.user_class_name,
         IFNULL(u.special_authorities, ''),
         u.status,
         u.no_password_indicator,
         IFNULL(u.previous_signon, '0001-01-01 00:00:00.000000'),
         IFNULL(u.text_description, ''),
         u.limit_capabilities,
         IFNULL(g.special_authorities, '')
  INTO :userinfo
  FROM qsys2.user_info u
  LEFT JOIN qsys2.user_info g
        ON g.authorization_name = u.group_profile_name
  WHERE u.authorization_name = :pusrprf
        AND ((u.special_authorities LIKE '%*ALLOBJ%'
              AND u.authorization_name NOT LIKE 'Q%')
             OR (IFNULL(g.special_authorities, '') LIKE
                 '%*ALLOBJ%' ));
```

```
textMessage = '{"text": "*[IBMi] '
+ %trim(SdsSystemName)
+ '* - User Profile '
+ %Trim(Pusrprf)
+ ' - '
+ %trim(userInfo.textDescription)
+ ' '
+ ' ('
+ %trim(userInfo.status)
+ ') '
+ ' was '
+ %Trim(Usage)
+ '"}';
```

```
EXEC SQL
```

```
VALUES
```

```
qsys2.http_post(TRIM(:url), TRIM(:textmessage)) INTO :resultatRequete;
```

```

Work with Exit Programs

Exit point:  QIBM_QSY_CHG_PROFILE      Format:  CHGP0100

Type options, press Enter.
  1=Add      4=Remove      5=Display      10=Replace

      Exit
      Program
Opt   Number      Program      Library
-----
      1
      2147483647  MSGUSR      SYSGIGC
      QGLDPUEXIT  QSYS
  
```

## Programmes d'exit

Nom du point d'exit ↑↓	Format de point d'exit ↑↓	Enregistré ↑↓	Terminé ↑↓	Numéro de programme d'exit ↑↓	Bibliothèque de programmes d'exit ↑↓	Programme d'exit ↑↓	Texte descriptif ↑↓	CCSID des ↑↓	Unités d'exécution multiples autorisées ↑↓	Action travail d'exécution n
QIBM_QSY_CHG_PROFILE	CHGP0100	YES	YES	1	SYSGIGC	MSGUSR	Envoi MSG si profil avec droits spéciaux	0		*MSG
QIBM_QSY_CHG_PROFILE	CHGP0100	YES	YES	2 147 483 647	QSYS	QGLDPUEXIT	IBM Tivoli Directory Server	297	YES	*MSG

Lignes sélectionnées: 1 | Lignes filtrées: 2 | Nombre total de lignes: 111



IBMi Application 12 juin, 11:24

[IBMi] GIGCTST - User Profile ██████ was changed

IBMi Application 12 juin, 19:06

[IBMi] GIGC - User Profile ARCAD - ARCAD objects owner (\*DISABLED) was created

IBMi Application 12 juin, 19:10

[IBMi] GIGC - User Profile ARCAD\_NET - ARCAD intersystem user (\*DISABLED) was created

[IBMi] GIGC - User Profile ARCAD - ARCAD objects owner (\*DISABLED) was changed

[IBMi] GIGC - User Profile ARCAD\_PGMR - ARCAD sample user profile (\*ENABLED) was created

[IBMi] GIGC - User Profile ARCAD\_PGMR - ARCAD sample user profile (\*ENABLED) was changed

```

**free
// =====
// Service..... :
// Application.. : Profils Utilisateur
// Objet..... : Envoi d'un message lors d'une opération sur les profils utilisateur *ALLOBJ
// Création..... : 05/05/22 par Sébastien BOURREAU
// Modifications :
// + / / par
// =====
// A positionner sur le point d'exit QIBM_QSY_CHG_PROFILE
// =====

```

```

/DEFINE CtlOpt_OPTION
Ctl-Opt OPTION(*SRCSTMT :*NODEBUGIO:*NOUNREF) EXPROPTS(*ALWBLANKNUM) DEBUG(*RETVAl);

```

```

dcl-pr MSGUSR ExtPgm('MSGUSR');
  NewUsrInfo char(38) const;
end-pr;

```

```

dcl-pi MSGUSR;
  NewUsrInfo char(38) const;
end-pi;

```

```

// Declare datastructure for breaking down the info
dcl-ds NewUsrSub;
  pExtPgm CHAR(20);
  pExtFmt CHAR(8);
  pUsrPrf CHAR(10);
end-ds;

```

```

dcl-s URL varchar(200) inz;
dcl-s resultatRequete varchar(100) inz;
dcl-s textMessage varchar(200) inz;
dcl-s usage char(10) inz;
dcl-s CmdString char(200);

```

```

dcl-ds userInfo qualified;
  authorizationName varchar(10);
  groupProfileName varchar(10);
  userClassName varchar(10);
  specialAuthorities varchar(88);
  status varchar(10);
  noPasswordIndicator char(3);
  previousSignon char(26);
  textDescription varchar(50);
  limitCapabilities char(10);
  groupSpecialAuthorities varchar(88);
end-ds;

```

```

EXEC SQL SET OPTION commit = *none,
          SRTSEQ = *LANGIDSHR,
          CLOSQLCSR = *ENDMOD;

```

```

// The CCSID 65535 is not supported
Callp(e) Execute('CHGJOB CCSID(1147)':18);

```

```

// Bring in the information
NewUsrSub = NewUsrInfo;

```

```

URL = 'https://chat.googleapis.com/' ;

```

```

Select;
When pExtFmt = 'CHGP0100';
  Usage = 'changed';
When pExtFmt = 'CRTP0100';
  Usage = 'created';
When pExtFmt = 'DLTP0100';
  Usage = 'deleted';
When pExtFmt = 'DLTP0200';
  Usage = 'deleted';
When pExtFmt = 'RSTP0200';
  Usage = 'restored';
Endsl;

```

```

clear userInfo;

```

```

// Is the profil as the special autoritie *ALLOBJ ?
EXEC SQL

```

```

  SELECT u.authorization_name,
         u.group_profile_name,
         u.user_class_name,
         IFNULL(u.special_authorities, ''),
         u.status,
         u.no_password_indicator,
         IFNULL(u.previous_signon, '0001-01-01 00:00:00.000000'),
         IFNULL(u.text_description, ''),
         u.limit_capabilities,
         IFNULL(g.special_authorities, '')
  INTO :userinfo
  FROM qsys2.user_info u
       LEFT JOIN qsys2.user_info g
         ON g.authorization_name = u.group_profile_name
 WHERE u.authorization_name = :pusrprf
       AND ((u.special_authorities LIKE '%*ALLOBJ%'
            AND u.authorization_name NOT LIKE 'Q%')
            OR (IFNULL(g.special_authorities, '') LIKE '%*ALLOBJ%'));

```

```

// If yes, send the message
if userInfo.authorizationName <> *blanks;

```

```

textMessage = '{"text": "'[IBMi] '
+ %trim(SdsSystemName)
+ '* - User Profile '
+ %Trim(Pusrprf)
+ ' - '
+ %trim(userInfo.textDescription)
+ ' '
+ ' ('
+ %trim(userInfo.status)
+ ') '
+ ' was '
+ %Trim(Usage)
+ ''}';

```

```

EXEC SQL VALUES qsys2.http_post(TRIM(:url), TRIM(:textmessage)) INTO :resultatRequete;
endif;

```

```

Callp(e) Execute('CHGJOB CCSID(*USRPRF)':21);

```

```

// Now we can end
Return;

```